

# 9/8

## Utilitaires

---

Ce chapitre est consacré aux « programmes-outils ». Ceux-ci peuvent améliorer les performances de votre AMSTRAD en lui ajoutant de nouvelles fonctions comme la copie d'écran graphique par exemple ou en vous facilitant la tâche dans divers domaines, programmation, gestion des disquettes, etc.

### 9/8.1

## Copie d'écran graphique

---

Ce programme s'adresse à tous les possesseurs d'imprimantes DMP-1, DMP-2000 ou équivalentes. Il permet de faire une copie d'écran graphique, en Basic.

Le principe de recopie d'écran est le suivant :

- 1) Initialisation de l'imprimante,
- 2) Définition du saut entre deux lignes,
- 3) Pour chaque ligne, grâce à la fonction TEST, les pixels allumés (d'une couleur autre que celle du fond de l'écran) sont mémorisés,
- 4) Les données mémorisées sont ensuite envoyées à l'imprimante.

```

3000 '-----
3010 '-----Initialisations-----
3020 '-----
3030 PRINT#8,CHR$(27);"a":PRINT#8,CHR$(27);"1"
3040 ORIGIN 0,0:DIM Z(320):VE=400:i=0
3050 '-----
3060 '-----Calcul d'une ligne graphique-----
3070 '-----
3080 FOR H=0 TO 639 STEP 2
3090   C#=""
3100   FOR X=0 TO 12 STEP 2
3110     IF TEST(H,VE-X)=0 THEN C#=C#+"0" ELSE C#=C#+"1"
3120   NEXT X
3130   I=I+1:Z(I)=VAL("&X"+C#)
3140 NEXT H
3150 GOSUB 3220 'Trace
3160 VE=VE-14
3170 I=0:IF VE>0 THEN 3080
3180 END
3190 '-----
3200 '-----Trace d'une ligne graphique-----
3210 '-----
3220 A=0:AF=0:A(0)=1:A(1)=100:A(2)=100:A(3)=120
3230 FOR J=1 TO 3
3240   A=A+A(J-1):AF=AF+A(J)
3250   PRINT#8,CHR$(27);"K";CHR$(A(J));CHR$(0);
3260   FOR K=A TO AF
3270     PRINT#8,CHR$(Z(K));
3280   NEXT K
3290 NEXT J
3300 PRINT#8
3310 RETURN

```

Lignes 3000 à 3040 : Initialisation de l'imprimante et définition de l'écart entre deux lignes

Lignes 3050 à 3180 : Identification et mémorisation des pixels « allumés » sur une ligne

Lignes 3190 à 3310 : Affichage d'une ligne graphique